
[bookmark: _Toc514234892]16.2.0	General
TTCN‑3 uses altsteps to specify default behaviour or to structure the alternatives of an alt statement.
Syntactical Structure
altstep [ interleave ] AltstepIdentifier
"(" [ { ( FormalValuePar | FormalTemplatePar ) [","] } ] ")"
[ runs on ComponentType ] 
[ mtc ComponentType ]
[ system ComponentType ]
"{" 
		{ ( VarInstance | TimerInstance | ConstDef | TemplateDef ) [";"] }
		AltGuardList
"}"

Semantic Description
Altsteps are scope units similar to functions. The altstep body defines an optional set of local definitions and a set of alternatives, the so-called top alternatives, that form the altstep body. The syntax rules of the top alternatives are identical to the syntax rules of the alternatives of alt statements. 
An altstep can also be used to define a named interleave statement by usage of the interleave keyword. In this case, the syntax rules of the top alternatives are identical to the syntax rules of the alternatives of the interleave statements.
NOTE 1:	As an interleave statement is semantically equivalent with an alt statement, there are no further restrictions on usages of interleave altsteps than on usages of normal altsteps. They can both be used as activated default alternatives and as top-level alternatives in other alt statement blocks.
The behaviour of an altstep can be defined by using the program statements and operations summarized in clause 18. Altsteps may invoke functions and altsteps or activate altsteps as defaults. 
Altsteps may be parameterized as defined in clause 5.4.
Altsteps may have an mtc clause. If an altstep has an mtc clause, the type referenced by this clause shall be mtc-compatible (see clause 6.3.3) with the type of the mtc component reference. If the mtc clause is not present, the type of the mtc component reference is unknown in the scope of this altstep. 
Altsteps may have a system clause. If an altstep has a system clause, the type referenced by this clause shall by system‑compatible (see clause 6.3.3) with the type of the system component reference. If the system clause is not present, the type of the system component reference is unknown in the scope of this altstep.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
a)	The local definitions of an altstep shall be defined before the set of alternatives. 
b)	The evaluation of formal parameters' default values and initialization of local definitions by calling value returning functions may have side effects. To avoid side effects that cause an inconsistency between the actual snapshot and the state of the component, and to prevent different results of subsequent evaluations on an unchanged snapshot, restrictions given in clause 16.1.4 shall apply to the formal parameters' default values and the initialization of local definitions.
c)	If an altstep includes port operations or uses component variables, constants or timers the associated component type shall be referenced using the runs on keywords in the altstep header. The one exception to this rule is if all ports, variables, constants and timers used within the altstep are passed in as parameters.
d)	An altstep without a runs on clause shall never invoke a function or altstep or activate an altstep as default with a runs on clause locally.
e)	An altstep that is activated as a default shall only have in value or template parameters. An altstep that is only invoked as an alternative in an alt statement or as stand-alone statement in a TTCN‑3 behaviour description may have in, out and inout parameters. The rules for formal parameter lists shall be followed as defined in clause 5.4.
f)	Altsteps started by using the start test component operation shall always have a runs on clause (see clause 22.5) and are considered to be invoked in the component to be started, i.e. not locally. However, the start test component operation may be invoked within behaviours without a runs on clause.
g)	If the altstep is an interleave alstep, all restrictions of the interleave statement (see clause 20.4) apply to the top alternatives of the altstep as well.
Examples
EXAMPLE 1:	Parameterized altstep with runs on clause
	// Given
	type component MyComponentType {
		var integer vc_myIntVar := 0;
		timer tc_myTimer;
		port myPortTypeOne pCO1, pCO2;
		port myPortTypeTwo pCO3;
	}

	// Altstep definition using pCO1, pCO2, vc_myIntVar and tc_myTimer of MyComponentType
	altstep a_altSet_A(in integer p_myPar1) runs on MyComponentType {
		[] pCO1.receive(mw_myTemplate(p_myPar1, vc_myIntVar)) {
				setverdict(inconc);
		   }
		[] pCO2.receive {
				if (p_myPar1 != 0) {
					repeat
				}
				else {
					break
				}
		   }
		[] tc_myTimer.timeout {
			setverdict(fail);
				stop
		   }
	}

EXAMPLE 2:	Altstep with local definitions
	altstep a_anotherAltStep(in integer p_myPar1) runs on MyComponentType {
		var integer v_myLocalVar := f_myFunction();			// local variable
		const float c_myFloat := 3.41;						// local constant
		[] pCO1.receive(MyTemplate(p_myPar1, v_myLocalVar)) {
			setverdict(inconc);
		   }
		[] pCO2.receive {
				repeat
		   }
	}
EXAMPLE 3:	Interleave Altstep 
	altstep interleave a_interleaveAltStep(in integer p_myPar1, in integer p_myPar2) 
		runs on MyComponentType {
		var integer v_myLocalVar := f_myFunction();			// local variable
		[] pCO1.receive(MyTemplate(p_myPar1, v_myLocalVar)) {}
		[] pCO1.receive(MyTemplate(p_myPar2, v_myLocalVar)) {}
	}

